// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/DutchAuction.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.17;

import '@openzeppelin/contracts/access/Ownable.sol';
import './IDutchAuction.sol';

/**
 * @title Dutch Auction contract made by Artiffine.
 * @author https://artiffine.com/
 */
contract DutchAuction is IDutchAuction, Ownable {
    uint256 public startTime; // Auction start time.
    uint256 public stepTime; // How many seconds is one auction round.
    uint256 public startPrice; // Start price of auction in wei.
    uint256 public priceDiscount; // Price discount every round in wei.
    uint256 public finalPrice; // Final price of auction in wei.

    error AuctionIsClosed();
    error StartTimeNotInTheFuture();
    error StepTimeIsZero();
    error StartPriceIsZero();
    error DiscountIsZero();
    error WrongAuctionSetup();

    constructor(uint256 _startTime, uint256 _stepTime, uint256 _startPrice, uint256 _priceDiscount, uint256 _finalPrice) {
        startAuction(_startTime, _stepTime, _startPrice, _priceDiscount, _finalPrice);
    }

    /**
     * @notice Returns current price auction in wei.
     *
     * @dev Throws `AuctionIsClosed()` error.
     */
    function getAuctionPrice() external view returns (uint256) {
        if (block.timestamp < startTime) revert AuctionIsClosed();
        uint256 maxSteps = (startPrice - finalPrice) / priceDiscount;
        uint256 timeElapsed = block.timestamp - startTime;
        uint256 currentStep = timeElapsed / stepTime;
        uint256 price;
        if (currentStep >= maxSteps) {
            price = finalPrice;
        } else {
            price = startPrice - (currentStep * priceDiscount);
        }
        assert(price >= finalPrice);
        return price;
    }

    /**
     * @notice Owner function to reset auction paramters.
     */
    function resetAuction(
        uint256 _startTime,
        uint256 _stepTime,
        uint256 _startPrice,
        uint256 _priceDiscount,
        uint256 _finalPrice
    ) external onlyOwner {
        startAuction(_startTime, _stepTime, _startPrice, _priceDiscount, _finalPrice);
    }

    function startAuction(
        uint256 _startTime,
        uint256 _stepTime,
        uint256 _startPrice,
        uint256 _priceDiscount,
        uint256 _finalPrice
    ) internal {
        if (_startTime <= block.timestamp) revert StartTimeNotInTheFuture();
        if (_stepTime == 0) revert StepTimeIsZero();
        if (_startPrice == 0) revert StartPriceIsZero();
        if (_priceDiscount == 0) revert DiscountIsZero();
        if (_startPrice - _priceDiscount < _finalPrice) revert WrongAuctionSetup();
        startTime = _startTime;
        stepTime = _stepTime;
        startPrice = _startPrice;
        priceDiscount = _priceDiscount;
        finalPrice = _finalPrice;
    }
}

// File: contracts/IDutchAuction.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.17;

/**
 * @title Dutch Auction interface made by Artiffine.
 * @author https://artiffine.com/
 */
interface IDutchAuction {
    /**
     * @notice Returns current price auction in wei.
     *
     * @dev Throws `AuctionIsClosed()` error.
     */
    function getAuctionPrice() external view returns (uint256);

    /**
     * @notice Owner function to reset auction paramters.
     */
    function resetAuction(
        uint256 _startTime,
        uint256 _stepTime,
        uint256 _startPrice,
        uint256 _priceDiscount,
        uint256 _finalPrice
    ) external;
}